A Concise Model for Static Induction Transistor IV Characteristics

Michael Rothacher

Introduction

Static Induction Transistors (SITs) are solid-state devices with triode-like characteristics. They are seen in high-frequency applications such as radar. They have also been used less often in audio frequency amplifiers and have been the subject of renewed interest for audio due to the efforts of manufacturers such as Digital Do Main, who produce modern versions of Yamaha parts from the 1970's, and First Watt, who have developed a silicon carbide SIT.

SITs are currently quite rare and expensive, and so we find it useful to simulate their action in computer software, such as SPICE and its many variants. This article describes a compact model for SIT drain current as a function of drain and gate voltages which can easily be incorporated into a SPICE subcircuit model. It appears to fit SIT curves more closely than using vacuum triode or MOSFET models and it contains just a few parameters, so it is not terribly difficult to fit data with trial and error methods.

Since SIT curves are generally more triode-like than pentode-like, my first attempts to make SIT models employed the triode equations of Norman Koren, which are widely accepted and popular for their ability to accurately fit tube characteristic curves. They actually worked quite well for modeling SITs, indeed close enough for most of the simple things I wanted to simulate. However, there were some areas in the curves that were difficult to fit well, such as the region where drain voltage approaches zero, and the area where drain current is very low. Also, many SIT curves I examined showed some increase in amplification factor with drain voltage. With the triode equations, it seemed I could often fit one or maybe two of these areas, but never all of them at once with good success.

Model Evolution

Before introducing the new model, it will be helpful to first examine how it evolved from a few of the key triode models.

We'll start with Equation (1) which represents an over-simplified model of a triode:

$$i_p = K\left(V_g + \frac{V_p}{\mu}\right) \tag{1}$$

The plate current is i_p, K is a constant originally called perveance, V_g is the grid voltage, V_p is the plate voltage, and *mu* is the amplification factor. In **Figure 1** below, I have attempted to use equation (1)



Figure 1 Power SIT curves based on equation (1)

to model a power static induction transistor. It's a rough and certainly optimistic approximation of the real thing.

In equation (2), an exponent is added:

$$i_p = K\left(V_g + \frac{V_p}{\mu}\right)^{\frac{3}{2}} \tag{2}$$

This equation is the basic triode model of Spangenberg (1948), Reynolds (1993), and Leach (1995). The 3/2 exponent comes from Child's Law, and is of course what gives each sweep its exponential shape. Again, I've used the equation to model a power SIT device, and our curves are now looking a bit more SIT-like (**Figure 2**).

Now if we make the exponent variable (equation (3)), we gain some additional control over the shape of our curve set (**Figure 3**).



Figure 2 Power SIT curves based on equation (2)

$$i_p = K\left(V_g + \frac{V_p}{\mu}\right)^X \quad (3)$$

Koren's model



Figure 3 Power SIT curves based on equation (3)

(equation (4)) does this, plus adds several new parameters to a pair of equations to help us really bend the numbers to our will. Koren's model fits a great many vacuum tubes with amazing accuracy.

$$E_1 = \frac{E_p}{k_p} \ln\left[1 + \exp\left(k_p\left(\frac{1}{\mu} + \frac{E_g + V_{ct}}{\sqrt{\left(k_{vb} + E_p^2\right)}}\right)\right)\right]$$

and

$$I_p = E_1^{Ex}/k_g(1 + sgn(E_1)) \tag{4}$$

Figure 4 is my original Koren model for the 2SK82. This is looking a lot more like a real SIT and it worked just fine for many basic simulations, but we can improve it a bit, and so we will.



Figure 4 2SK82 curves based on Koren's model.

The Concise SIT Equation

All of this brings us to the new concise model for SITs according to equation (5) below. It wasn't really practical to pursue a new model based on physical device characteristics, since so little physical data is available for most SITs. Therefore, the concise model, like Koren's model is phenomenological or behavioral and none of the parameters are (necessarily) related to the physical properties or internal geometry of the device being modeled.

$$i_d = K\left(V_g + N[\ln(V_d)] + \frac{Vd}{\mu}\right)^X \tag{5}$$

Parameters K, X, and mu are similar to the parameters in equation (3) and I've kept their designations the same as such. The new term here is the natural logarithm of the drain voltage multiplied by a coefficient N. This expression helps take care of the trouble areas mentioned earlier. The N parameter can be thought of as a pentode-triode shape control. And it is important in any of these models not to take the mu parameter too literally. Sometimes the chosen mu will correspond well with the datasheet figure and sometimes it will need to be very different in order to achieve the proper fit.

One way to determine the parameter values is to extract X,Y data from a set of IV curves and run them through curve-fitting software such as MathCAD. However, it is entirely possible to achieve a good fit using a trial and error method whereby the parameters are tweaked while a set of curves is "painted" over an image taken from a datasheet or some other source. A number of vacuum tube modeling packages work in this manner, and I have written some software, presented later in this article, which is freely available for you to use. All of the models in this article were made using this method, generally requiring thirty minutes to an hour's effort to arrive at something reasonable.

We'll need to tart this up a bit to get it ready for SPICE. The *uramp(x)* function in equation (6) is a SPICE-specific function which returns x when x is greater than zero and returns zero otherwise. If your version of SPICE doesn't support *uramp(x)* you can try using the *max(x,y)* function or the *limit(x,y,z)* function. Examples of these alternatives are presented with the SPICE model sample below. I've also added *Vct*, which is similar to *Vct* in Koren's model. It is a *Vgs* shift. It isn't strictly necessary, but can be useful in fitting, and is also useful, when parameterized, for modeling a particular sample.

Figure 5 is a set of curves for the Sony 2SK82 produced with equation (5).



Figure 5 2SK82 curves based on eq. (5).

Figure 6 is the same set of curves overlaid onto 2SK82 curve tracer output from a Sony document, and it appears we have achieved a very nice fit.



Figure 6 2SK82 curves based on eq. (5) overlaid on curve tracer output.

SPICE Verification

It is now a fairly straightforward task to incorporate the new equation into a SPICE subcircuit model. The simplest approach is to use a behavioral source:

```
.SUBCKT 2SK82 D G S

+ PARAMS: K=0.157 MU=7.3 N=2.24 X=1.51 VCT=0.0

B1 D S I=K*PWR(URAMP((V(G,S)+VCT)+(N*LN(V(D,S))+(V(D,S)/MU))),X)

.ENDS 2SK82
```

Here are examples of alternatives to *uramp(x):*

```
B1 D S I=K*PWR(MAX((V(G,S)+VCT)+(N*LN(V(D,S))+(V(D,S)/MU)),0),X)
```

or:

```
B1 D S I=K*PWR(LIMIT((V(G,S)+VCT)+(N*LN(V(D,S))+(V(D,S)/MU)),0,1.0E6),X)
```

Now, let's build up a SPICE deck and plot some IV curves in SPICE:

```
EXAMPLE OUTPUT CHARACTERISTICS OF 2SK82 MODEL

VGS 1 0 DC 0V

VX 3 2 DC 0V

VDD 3 0 DC 55V

XJ1 2 1 0 2SK82

.SUBCKT 2SK82 D G S

+ PARAMS: K=0.157 MU=7.3 N=2.24 X=1.51 VCT=0.0

B1 D S I=K*PWR(URAMP((V(G,S)+VCT)+(N*LN(V(D,S))+(V(D,S)/MU))),X)

*1 MEG GATE-SOURCE RESISTOR FOR SIMULATION

R1 G S 1MEG

.ENDS 2SK82

.SAVE ALL VGS
```

```
.DC VDD 50 0 -0.2 VGS 4 -18 -2

.PLOT DC I(VX)

.END
```

Here, the voltage source VDD is swept from 0-50 volts, and another voltage source VGS is swept from 4 to -18 volts in -2 volt steps. VX is simply a 0V source for capturing the drain current. **Figure 7** is our plotted output from SPICE.



Figure 7 2SK82 curves from SPICE based on eq. 5.

Next, we'll test it in a common source amplifier circuit. **Figure 8** is the circuit we'll simulate. You may notice I've added some inter-electrode capacitances to the model and assigned some values. These values are merely rough estimates I made for a specific operating region by measuring the high frequency response of a real common source stage, calculating the effective input capacitance, and splitting the value symmetrically. As modeled, these capacitances are static, whereas their real-world counterparts are a non-linear function of voltage, typically with input capacitance decreasing as voltage increases. You can include theses capacitances in your own models if you wish, or leave them out depending on what you're interested in modeling.



Figure 8 2SK82 SPICE common source test jig.

Here is the SPICE deck for Figure 8:

```
EXAMPLE 2SK82 CS AMPLIFIER

V2 1 0 DC 50V

V1 4 0 AC 1V

XJ1 2 4 3 2SK82

RD 1 2 13

RS 3 0 1

RG 4 0 47K

C1 5 2 1000U

RL 5 0 8

.SUBCKT 2SK82 D G S ; Drain Gate Source

+ PARAMS: MU=7.3 X=1.51 K=0.157 N=2.24 VCT=0 RG=2MEG

*______________________________________________

B1 D S I=K*PWR(URAMP((V(G,S)+VCT)+(N*LN(V(D,S))+(V(D,S)/MU))),X)

R1 G S {RG}
```

```
CGS G S 1000P
CGD G D 1000P
CDS G S 0P
.ENDS 2SK82
.OP
.END
```

First, we'll look at the DC operating point analysis (.OP):

```
DC Operating Point ...
v(1) = 5.000000e+01
v(2) = 1.874658e+01
v(3) = 2.404108e+00
v(4) = 0.000000e+00
v(5) = 0.000000e+00
v1#branch = 1.202054e-06
v2#branch = -2.40411e+00
```

The idle current is 2.4A and the SIT drain is at 18.7V.

Next, we'll run an AC analysis with a 600 Ohm source. In **Figure 9**, you can see the effect of the inter-electrode capacitances, and the output capacitor.



Figure 9 Frequency response of figure 8 test circuit.

And, just for good measure, here's a Fourier analysis. THD is around .100%. The second harmonic is dominant, followed by a bit of third with things getting vanishingly small from there. All of this agrees reasonably well with what I've seen from real samples, in real circuits. The gain from the Sony curve tracer output is slightly lower than some of my real SIT samples, but there is some gain variation expected for these parts, and it would be easy enough to tweak the model for this if needed. It's certainly good enough to try this model in various topologies to see what happens.

Fourier analysis for v(5):

No. Harmonics: 10, THD: 0.100356 %, Gridsize: 200, Interpolation D

Harmonic Frequency Magnitude Phase Norm. Mag Norm. P

——— ———- ———- ——- ———- ——-

0 0.000000e+00 2.150954e-03 0.000000e+00 0.000000e+00 0.00000

1 1.000000e+03 1.133021e+00 1.794477e+02 1.000000e+00 0.00000

2 2.000000e+03 1.064998e-03 -9.17869e+01 9.399628e-04 -2.7123

3 3.000000e+03 2.342712e-04 -1.48347e+02 2.067668e-04 -3.2779

4 4.000000e+03 1.308045e-04 -8.98848e+01 1.154475e-04 -2.6933

5 5.000000e+03 1.317666e-04 -9.01006e+01 1.162967e-04 -2.6954

6	6.000000e+03 1.317159e-04 -9.03272e+01 1.162520e-04 -2.6977
7	7.000000e+03 1.316647e-04 -9.04942e+01 1.162068e-04 -2.6994
8	8.000000e+03 1.316059e-04 -9.06475e+01 1.161549e-04 -2.7009
9	9.000000e+03 1.315397e-04 -9.07913e+01 1.160964e-04 -2.7023

Interactive Modeling Software

I have written a software application which you can use to generate models of your own, or just fiddle with the various parameters to see how they alter the curves. The application requires Google's Chrome browser to function properly. **Figure 10** shows a representative screen shot.

The app is located at https://blog.audiomaker.tech/sit-modeller/. You control the graph by using the up/down arrows to change parameter values, or you can enter values directly. You can change the axis range settings to suit your needs. The axis range settings will be stored in a comment line in the subcircuit model. This is to remind you that the model may only be valid for the specified range/scale, though it may be valid outside a given range depending on the device being modeled and the accuracy of fit. If desired, you can also select a reference image which will be displayed in the background. This image will need to be pre-trimmed in MS Paint or other image editing software so that it only includes the plot area of the graph. Click on "Choose File", browse to your image and it will be loaded and scaled to fit the current plot window. The "Image Opacity" setting controls the opacity of the reference image to make it easier to see the reference graph against the model curves.

If you click in the graph area, the drain resistance and amplification factor are calculated and displayed for you. If you know the interelectrode capacitances, you can enter them. The SPICE model is updated whenever you make a change, and once you have things the way you want them, you can copy/paste the text into your SPICE file.



Figure 10 Representative screen shot of the author's modeling application.

Additional Examples

Figure 11 shows a few additional outputs from the modeling application for several devices, along with parameter values (**Table 1**) which you may use as a starting basis for your own models:



2SK60



2SK77



2SK79



2SK63

Figure 11 Modeling application output curves for several SIT devices.

Device	K	Mu	N	X	Vct
2SK82	0.157	7.3	2.24	1.51	0
2SK60	0.06	18	2.7	1.57	0
2SK77	0.78	15.4	2.22	1.29	0
2SK79	0.0023	33.15	0.096	3.97	0.62
2SK63	0.0017	30.6	0.14	4.4	0.47

Table 1 Suitable parameter start values for the modeling results from figure 11.

"The Map Is Not the Territory"

I've borrowed this section title from Nelson Pass, who once used Korzybski's dictum to issue a kindly word of caution in the art of circuit simulation. It is especially apropos here. It is first important to understand that our models can only hope to *approximate* reality. Moreover, we must remember that the information we use to build our models may be flawed, for example, most datasheet curves from the nineties or earlier were hand-drawn by draftsmen who attempted to draw smooth curves through measured data, so it is certainly possible that some subjective information was introduced. Even when photographs of curve tracer output are used to build models, remember that those curves represent just one specific sample of a given part, under specific conditions, and so forth, and of course, we know that there can be a good bit of variation in parameters among samples of the same part number. My point is simply that you shouldn't lose too much sleep over these things. Eventually, you will have to build your circuits with real parts and real electricity anyway, and things are sure to behave a little differently. The idea is to build models that work *well enough* to give you some insight into how the real thing will work in your circuit.

The concise model is capable of producing results which are quite good in simulation. As it is, it doesn't model temperature effects, non-linear capacitances and so forth; in that respect it is much like most vacuum tube models. It is compact enough for trial and error fitting, and it is easily implemented in many versions of SPICE. I hope you might find this model useful for one of your own projects, or at the very least fun to study and play around with in SPICE. Just remember, you can't listen to a simulator.

Now go build something.

References and Additional Information

1. An Introduction to Static Induction Transistors by Nelson Pass (www.firstwatt.com)

2. Norman Koren Vacuum Tube Audio Page at www.normankoren.com/audio

3. WinSpice was used to generate output for this article, and the model has been tested with Multisim and, of course, the DIY favorite, LTspice.